Infinite Potential Wall

The energy eigenfunctions and eigenvalues of the infinite potetial wall:

```
ψ[x_, n_] = Sqrt[2/L] Sin[n π/L x]; En[n_] = ħ^2 π^2/(2 m L^2) n^2;
```

```
$Assumptions = n ∈ Integers && l ∈ Integers && n > 0 && m > 0 && L > 0 && ħ > 0 && l > 0
```

```
n ∈ Integers && l ∈ Integers && n > 0 && m > 0 && L > 0 && ħ > 0 && l > 0
```

Problem 2:

Consider the particle in the initial state $\phi(x, 0) = A \sin^3\left(\frac{\pi}{L} x\right)$. Compute the state $\phi(x, t)$ and the corresponding average values of position , momentum etc.

Solution:

Consider the expression of the wave function interms of the energy eigenfunc– tions:

$$\phi(x, 0) \equiv \phi(x) = \sum_{n=1}^{\infty} c_n \psi_n(x), \quad c_n = \int_0^L \psi_n^*(x)\, \phi(x)\, dx$$

$$\phi(x, t) = \sum_{n=1}^{\infty} c_n \psi_n(x)\, e^{-i E_n t/\hbar}$$

$$< x > (t) = \sum_{n,m=1}^{\infty} c_m^* c_n e^{-i (E_n - E_m) t/\hbar} \int_0^L \psi_m^*(x)\, x\, \psi_n(x)\, dx$$

$$< x^2 > (t) = \sum_{n,m=1}^{\infty} c_m^* c_n e^{-i (E_n - E_m) t/\hbar} \int_0^L \psi_m^*(x)\, x^2\, \psi_n(x)\, dx$$

We compute the normalization factor A and define the wave function at t=0

```
A = 1/Sqrt[∫_0^L (Sin[π/L x])^6 dx]
```

$$\frac{4}{\sqrt{5}\,\sqrt{L}}$$

```
ϕ[x_, 0] = A Sin[π/L x]^3; ∫_0^L ϕ[x, 0]^2 dx
```

1

```
cn[n_] = ∫_0^L ψ[x, n] ϕ[x, 0] dx
```

$$\frac{24 \sqrt{\frac{2}{5}} \sqrt{\frac{1}{L}} \sqrt{L}\, \mathrm{Sin}[n \pi]}{\left(9 - 10 n^2 + n^4\right) \pi}$$

```
cn[n_] = ∫_0^L ψ[x, n] ϕ[x, 0] ⅆx /. Sin[n π] → 0
```

```
0
```

All coefficients are zero except when the denominator is 0:

```
Solve[9 - 10 n^2 + n^4 == 0, n]
```

```
{{n → -3}, {n → -1}, {n → 1}, {n → 3}}
```

```
cn[1] = ∫_0^L ψ[x, 1] ϕ[x, 0] ⅆx
```

$$\frac{3}{\sqrt{10}}$$

```
cn[3] = ∫_0^L ψ[x, 3] ϕ[x, 0] ⅆx
```

$$-\frac{1}{\sqrt{10}}$$

```
ϕ[x_, t_] = Sum[cn[n] ψ[x, n] ⅇ^(-ⅈ En[n] t/ℏ), {n, 1, 3}]
```

$$\frac{3\, e^{-\frac{i \pi^2 t \hbar}{2 L^2 m}} \sqrt{\frac{1}{L}}\ \mathrm{Sin}\left[\frac{\pi x}{L}\right]}{\sqrt{5}} - \frac{e^{-\frac{9 i \pi^2 t \hbar}{2 L^2 m}} \sqrt{\frac{1}{L}}\ \mathrm{Sin}\left[\frac{3 \pi x}{L}\right]}{\sqrt{5}}$$

```
Abs[{ϕ[x, 0], ϕ[x, 0.3], ϕ[x, 0.5]}] /. {x → 0.4, L → 1, m → 1, ℏ → 1}
```

```
{1.53884, 1.48333, 1.45589}
```

```
Absϕ[x_, t_] = Abs[ϕ[x, t]] /. {L → 1, m → 1, ℏ → 1};
Plot[{Absϕ[x, 0], Absϕ[x, 0.3], Absϕ[x, 0.5], Absϕ[x, 0.7], Absϕ[x, 1.]},
 {x, 0, 1}, PlotStyle → {Red, Green, Blue, Yellow, Magenta}]
```



Here you can see an animation of the time evolution of $|\phi(x, t)|^2$

```
Manipulate[Plot[Absϕ[x, t], {x, 0, 1}, PlotRange → {{0, 1}, {0, 2}}], {t, 0, 4, 0.0001}]
```



as well as the real and imaginary parts of the wave function $\text{Re}(\phi(x, t))$, $\text{Im}(\phi(x, t))$:

```
Reϕ[x_, t_] = Re[ϕ[x, t]] /. {L → 1, m → 1, ℏ → 1};
Imϕ[x_, t_] = Im[ϕ[x, t]] /. {L → 1, m → 1, ℏ → 1};
Manipulate[Plot[{Reϕ[x, t], Imϕ[x, t]}, {x, 0, 1},
  PlotRange → {{0, 1}, {-2, 2}}, PlotStyle → {Red, Blue}], {t, 0, 4, 0.0001}]
```



Average energy $<E>$, $<E^2>$:

```
avE = Sum[cn[n]^2 En[n], {n, 1, 3}]
```

$$\frac{9\,\pi^2\,\hbar^2}{10\,L^2\,m}$$

```
avE / En[1]
```

$$\frac{9}{5}$$

```
avE2 = Sum[cn[n]^2 En[n]^2, {n, 1, 3}]
```

$$\frac{9\,\pi^4\,\hbar^4}{4\,L^4\,m^2}$$

```
avE2 / En[1]^2
```

9

```
Δ E = Sqrt[avE2 - avE^2]
```

$$\frac{6}{5}\pi^2\sqrt{\frac{\hbar^4}{L^4\,m^2}}$$

```
Δ E/ En[1] // Simplify
```

$$\frac{12}{5}$$

```
xnm[n_, l_] = ∫_0^L ψ[x, n] x ψ[x, l] dx
```

$$\frac{4\left(-1+(-1)^{l+n}\right) l\,L\,n}{\left(l^2-n^2\right)^2\pi^2}$$

The $n = m$ case must be treated differently:

```
xnm[n_, n_] = ∫_0^L ψ[x, n] x ψ[x, n] dx
```

$$\frac{L}{2}$$

```
x2nm[n_, l_] = ∫_0^L ψ[x, n] x^2 ψ[x, l] dx
```

$$\frac{8\,(-1)^{l+n}\,l\,L^2\,n}{\left(l^2-n^2\right)^2\pi^2}$$

```
x2nm[n_, n_] = ∫_0^L ψ[x, n] x^2 ψ[x, n] dx
```

$$\frac{1}{6}L^2\left(2-\frac{3}{n^2\pi^2}\right)$$

$<x>(t) = <x>(0) = L/2$

$$\mathbf{xav\phi[t\_]} = \sum_{n=1}^{3}\sum_{l=1}^{3} \mathbf{cn[n]\ cn[l]\ xnm[n,\ l]}\ e^{-i\,(\mathrm{En}[l]-\mathrm{En}[n])\,t/\hbar}$$

$$\frac{L}{2}$$

$$\mathbf{x2av\phi[t\_]} = \sum_{n=1}^{3}\sum_{l=1}^{3} \mathbf{cn[n]\ cn[l]\ x2nm[n,\ l]}\ e^{-i\,(\mathrm{En}[l]-\mathrm{En}[n])\,t/\hbar}\ \mathbf{//\ ExpToTrig}$$

$$\frac{L^2}{3} - \frac{41\,L^2}{90\,\pi^2} - \frac{9\,L^2\,\mathrm{Cos}\left[\frac{4\,\pi^2\,t\,\hbar}{L^2\,m}\right]}{40\,\pi^2}$$

```
Plot[{x2avϕ[t]} /. {L → 1, m → 1, ℏ → 1}, {t, 0, 1}]
```



Momenta now:

$$\mathbf{pnm[n\_,\ l\_]} = \int_0^L \psi[x,\ n]\ (-i\,\hbar)\ \partial_x \psi[x,\ l]\ dx\ \ /.\ \ \mathrm{Cos}[l\,\pi] \to (-1)^l$$

$$\frac{2\,i\,\left(-1 + (-1)^{l+n}\right)\,l\,n\,\hbar}{L\,\left(-l^2 + n^2\right)}$$

$$\mathbf{pnm[n\_,\ n\_]} = \int_0^L \psi[x,\ n]\ (-i\,\hbar)\ \partial_x \psi[x,\ n]\ dx$$

$$0$$

$$\texttt{p2nm[n\_, l\_]} = \int_0^L \psi[x, n] \left(-\hbar^2\right) \partial_{x x} \psi[x, l] \, dx$$

0

$$\texttt{p2nm[n\_, n\_]} = \int_0^L \psi[x, n] \left(-\hbar^2\right) \partial_{x x} \psi[x, n] \, dx$$

$$\frac{n^2 \pi^2 \hbar^2}{L^2}$$

$$\texttt{pav}\phi\texttt{[t\_]} = \sum_{n=1}^{3} \sum_{l=1}^{3} \texttt{cn[n] cn[l] pnm[n, l]} \, e^{-i (\texttt{En[l]}-\texttt{En[n]}) t/\hbar}$$

0

$$\texttt{p2av}\phi\texttt{[t\_]} = \sum_{n=1}^{3} \sum_{l=1}^{3} \texttt{cn[n] cn[l] p2nm[n, l]} \, e^{-i (\texttt{En[l]}-\texttt{En[n]}) t/\hbar}$$

$$\frac{9 \pi^2 \hbar^2}{5 L^2}$$

Compare with the exact formula for $< p^2 > (0)$:

$$\int_0^L \phi[x, 0] \left(-\hbar^2\right) \partial_{x x} \phi[x, 0] \, dx$$

$$\frac{9 \pi^2 \hbar^2}{5 L^2}$$